where $S(x) = \pi R^2$ is the cross-sectional area distribution and $R(x) \ll L$, the body length.

15. Find the velocity distribution on the slender ellipsoid of revolution of thickness ratio τ, for which

$$R^2(x) = \tau^2(1 - x^2) \qquad -1 \le x \le 1$$

Warning: In computing the x component of velocity, do *not* try to simplify the integrand beyond using the result of problem 14 for the source strength.

16. From problem 15, if not 14, demonstrate that on the body surface, the x component of velocity induced by the sources is proportional to τ^2, where τ is the thickness ratio of the body, and the r component is proportional to τ. Because this behavior differs from the two-dimensional case, equation 3-35 is not the appropriate approximation to Bernoulli's equation in computing the pressure distribution on a slender body of revolution. How much equation 3-35 be modified?

17. Show that

$$W = \frac{1}{2\pi}(Q + i\Gamma)\ln z$$

is the complex potential of a source of strength Q and vortex of strength Γ at the origin.

18. Plot the exact pressure distribution on an ellipse of thickness ratio $\tau = 0.1$. Compare with thin-airfoil theory and the results of program DUBLET.

19. Show that our result for the complex potential of uniform flow past an ellipse reduces to its proper limit when $\tau = 1$ (so that the ellipse becomes a circle).

20. Use equations 3-30 and 3-82 to find the exact strength of the line source distribution associated with flow past an ellipse.

21. Find the shape of and pressure distribution on the Joukowsky airfoil defined by $\varepsilon/b = -0.1$. Program JOUKOW, listed under Section 3.12, is at your disposal. Note that the program's EPS is the text's ε/b. Use the output as input to program DUBLET, and check out the latter's performance. The results of JOUKOW are, of course, exact.

22. Revise DUBLET to provide a numerical solution to the problem of axisymmetric flow past a body of revolution. According to problem 11 of Chapter 2, the Stokes stream function of such a flow should be constant on the body, whereas from problems 12 and 13 of Chapter 2, the Stokes stream function of an axial distribution of point sources in a uniform flow is

$$\psi = \tfrac{1}{2}V_\infty r^2 - \int_{x_s}^{x} q(t)\frac{x - t}{[(x - t)^2 + r^2]^{1/2}}\,dt$$

Here $q(t)$ is the strength of the source distribution, per unit length, which should therefore be determined so that ψ is constant (zero for a closed body) at $r = R(x)$, the body contour.

3.12. COMPUTER PROGRAMS

```
C      PROGRAM RANKIN
C
C                     LOCATE POINT ON RANKINE OVAL
C
       PRINT,         ' DIMENSIONLESS SOURCE STRENGTH =',
       READ,          Q
       PRINT 1000
       XO             =  - Q + SQRT(Q*Q + 1.)
   100 PRINT,          ' Y/A =',
       READ,          Y
       X2             =  XO*XO - Y*Y + 2.*Y*XO/TAN(Y/Q)
       IF (X2 .GE. 0.0)      GO TO 110
       PRINT,         ' Y IS TOO LARGE'
       GO TO 100
   110 X              =  SQRT(X2)
       PSI            =  Y + Q*ATAN2(Y,X+XO) - Q*ATAN2(Y,X-XO)
C
C                     COMPUTE PSI, VELOCITY, AND PRESSURE AT POINT
C
       RPLUS          =  (X+XO)**2 + Y*Y
       RMINUS         =  (X-XO)**2 + Y*Y
       U              =  1. + Q*(X+XO)/RPLUS - Q*(X-XO)/RMINUS
       V              =  Q*Y/RPLUS - Q*Y/RMINUS
       CP             =  1.0 - U*U - V*V
       PRINT 1010, X,PSI,U,V,CP
       GO TO 100
  1000 FORMAT(/,20X,'X/A     PSI      U      V      CP')
  1010 FORMAT(15X,5F8.3)
       END

C      PROGRAM DUBLET
C
C                     INCOMPRESSIBLE AERODYNAMICS OF SYMMETRIC AIRFOIL
C                     AT ZERO ANGLE OF ATTACK BY LINE DOUBLET DISTRIBUTION
C
       COMMON         T(100),M(100),N,XS,XF
       REAL           M,MPLOT
C
C                     INPUT NUMBER OF INTERVALS N
C
       PRINT,         ' N =',
       READ,          N
C
C                     DETERMINE ENDPOINTS OF DISTRIBUTION XS, XF
C
   100 PRINT,         ' XS,XF =',
       READ,          XS,XF
       CALL FINDM
       CALL PRESS(0.0,U0,CP0)
       CALL PRESS(1.0,U1,CP1)
       PRINT,         ' U AT X = 0,1 =',U0,U1
       PRINT,         ' DO YOU ACCEPT THESE RESULTS (Y/N)',
       READ 1000, IANS
       IF (IANS .NE. 1HY)     GO TO 100
```

```
C
C                 OUTPUT RESULTS
C
      PRINT 1010
      M(N+1)      =  0.0
      DO 200      I = 1,N+1
      MPLOT       =  M(I)*3.1415926585
  200 CALL PLOTXY(T(I),MPLOT,100.)
      PRINT 1020
      DO 210      I = 1,N
      XX          =  .5*(T(I) + T(I+1))
      YY          =  Y(XX)
  210 CALL PLOTXY(XX,YY,100.)
      PRINT 1030
      READ,       NPRINT
      DO 220      I = 1,NPRINT
      XX          =  (I - 1)/FLOAT(NPRINT - 1)
      CALL PRESS(XX,U,CP)
  220 CALL PLOTXY(XX,CP,40.)
 1000 FORMAT(A1)
 1010 FORMAT(/,' DOUBLET STRENGTH DISTRIBUTION',/
     +       ' M = M(I) FOR T(I) < T < T(I+1)',//
     +       4X,'T(I)',5X,'M(I)/2',/)
 1020 FORMAT(//,' BODY SHAPE',//,4X,'X',9X,'Y',/)
 1030 FORMAT(//,' BODY SURFACE PRESSURE DISTRIBUTION',//,
     +       4X,'X',8X,'CP',//,' INPUT NUMBER OF OUTPUT POINTS',)
      STOP
      END
C=============================================================
      SUBROUTINE FINDM
C
C                 FIND DOUBLET STRENGTH TO MEET
C                 FLOW TANGENCY CONDITION
C
      COMMON      T(100),M(100),N,XS,XF

      COMMON /COF/ A(101,111),NEQNS
      REAL M
      PI          =  3.1415926585
      NP          =  N + 1
      DO 100      I = 1,NP
      FRACT       =  .5*(1. - COS(PI*(I-1)/FLOAT(N)))
  100 T(I)        =  XS + (XF - XS)*FRACT
C
C                 SET UP LINEAR SYSTEM OF EQUATIONS
C
      DO 210      I = 1,N
      XI          =  .5*(T(I) + T(I+1))
      YI          =  Y(XI)
      FAC1        =  ATAN2(T(1) - XI,YI)
      DO 200      J = 1,N
      FAC2        =  ATAN2(T(J+1) - XI,YI)
      A(I,J)      =  (FAC2 - FAC1)/YI
  200 FAC1        =  FAC2
  210 A(I,NP)     =  1.0
```



```
C
C                 SOLVE FOR DOUBLET STRENGTH
C
      NEQNS       =  N
      CALL GAUSS(1)
      DO 300      I = 1,N
  300 M(I)        =  A(I,NP)
      RETURN
      END
C=============================================================
      SUBROUTINE PRESS(X,U,CP)
C
C                 FIND PRESSURE COEFFICIENT CP AT (X,Y(X))
C
      COMMON      T(100),M(100),N,XS,XF
      REAL        M
      YB          =  Y(X)
      U           =  1.0
      V           =  0.0
      VF1         =  1./((T(1) - X)**2 + YB*YB)
      UF1         =  (T(1) - X)*VF1
      DO 100      J = 1,N
      VF2         =  1./((T(J+1) - X)**2 + YB*YB)
      UF2         =  (T(J+1) - X)*VF2
      U           =  U + M(J)*(UF2 - UF1)
      V           =  V - M(J)*YB*(VF2 - VF1)
      VF1         =  VF2
  100 UF1         =  UF2
      CP          =  1.0 - U*U - V*V
      RETURN
      END
C=============================================================
      FUNCTION Y(X)
C
C                 ORDINATE OF BODY CONTOUR
C
C                 EXAMPLE GIVEN IS ELLIPSE OF THICKNESS RATIO 0.1
C
      Y           =  0.1*SQRT(X*(1. - X))
      RETURN
      END
C=============================================================
      SUBROUTINE PLOTXY(X,Y,YMULT)
C
C                 PLOT Y ON SAME LINE AS X AND Y ARE PRINTED
C
      COMMON /SKAL/ NZERO,YMULT
      YSLOP       =  .5/YMULT
      NPLOT       =  (Y + YSLOP)*YMULT
      IF (Y + YSLOP .LT. 0.0)    NPLOT = NPLOT - 1
      IF (NPLOT) 10,20,30
C
C                 -- NEGATIVE Y
C
   10 NTOX        =  NZERO + NPLOT
      IF (NTOX .LT. 1)        GO TO 40
      NTODOT      =  - NPLOT - 1
      IF (NTODOT .EQ. 0)      GO TO 15
      PRINT 1010,             X,Y,NTOX,NTODOT
      RETURN
```

```
   15 PRINT 1015,            X,Y,NTOX
      RETURN
C
C                  --  ZERO Y
C
   20 PRINT 1020,            X,Y,NZERO
      RETURN
C                  --  POSITIVE Y
C
   30 NTOX       =  NPLOT - 1
      IF (NTOX + NZERO .GT. 60)          GO TO 40
      IF (NTOX .EQ. 0)                   GO TO 35
      PRINT 1030,            X,Y,NZERO,NTOX
      RETURN
   35 PRINT 1035,            X,Y,NZERO
      RETURN
C
C                  --  Y OUT OF RANGE OF PLOT
C
   40 PRINT 1040,            X,Y
      RETURN
 1010 FORMAT(F8.4,F10.4,=X,1HX,=X,1H.)
 1015 FORMAT(F8.4,F10.4,=X,2HX.)
 1020 FORMAT(F8.4,F10.4,=X,1HX)
 1030 FORMAT(F8.4,F10.4,=X,1H.,=X,1HX)
 1035 FORMAT(F8.4,F10.4,=X,2H.X)
 1040 FORMAT(F8.4,F10.4)
      END
C=========================================================
      SUBROUTINE GAUSS(NRHS)
C
C        SOLUTION OF LINEAR ALGEBRAIC SYSTEM BY
C        GAUSS ELIMINATION WITH PARTIAL PIVOTING
C
C                  [A]       =  COEFFICIENT MATRIX
C                  NEQNS     =  NUMBER OF EQUATIONS
C                  NRHS      =  NUMBER OF RIGHT-HAND SIDES
C
C                  RIGHT-HAND SIDES AND SOLUTIONS STORED IN
C                  COLUMNS NEQNS+1 THRU NEQNS+NRHS OF [A]
C
      COMMON /COF/ A(101,111),NEQNS
      NP         =  NEQNS + 1
      NTOT       =  NEQNS + NRHS
C
C                   GAUSS REDUCTION
C
      DO 150     I = 2,NEQNS
C
C                   --  SEARCH FOR LARGEST ENTRY IN (I-1)TH COLUMN
C                       ON OR BELOW MAIN DIAGONAL
C
      IM         =  I - 1
      IMAX       =  IM
      AMAX       =  ABS(A(IM,IM))
      DO 110     J = I,NEQNS
      IF (AMAX .GE. ABS(A(J,IM)))        GO TO 110
      IMAX       =  J
      AMAX       =  ABS(A(J,IM))
  110 CONTINUE
C
C                   --  SWITCH (I-1)TH AND IMAXTH EQUATIONS
C
      IF (IMAX .NE. IM)      GO TO 140
      DO 130      J = IM,NTOT
      TEMP        =  A(IM,J)
      A(IM,J)     =  A(IMAX,J)
      A(IMAX,J)   =  TEMP
  130 CONTINUE
C
C                   ELIMINATE (I-1)TH UNKNOWN FROM
C                   ITH THRU (NEQNS)TH EQUATIONS
C
  140 DO 150      J = I,NEQNS
      R           =  A(J,IM)/A(IM,IM)
      DO 150      K = I,NTOT
  150 A(J,K)      =  A(J,K) - R*A(IM,K)
C
C                   BACK SUBSTITUTION
C
      DO 220      K = NP,NTOT
      A(NEQNS,K)  =  A(NEQNS,K)/A(NEQNS,NEQNS)
      DO 210      L = 2,NEQNS
      I           =  NEQNS + 1 - L
      IP          =  I + 1
      DO 200      J = IP,NEQNS
  200 A(I,K)      =  A(I,K) - A(I,J)*A(J,K)
  210 A(I,K)      =  A(I,K)/A(I,I)
  220 CONTINUE
      RETURN
      END
C     PROGRAM JOUKOW
C
C        FIND SHAPE OF AND PRESSURE DISTRIBUTION ON
C        SYMMETRIC JOUKOWSKI AIRFOIL AT ZERO ANGLE OF ATTACK
C
      COMPLEX     Z,ZTILDE,ZPRIME,WPRIME,I
      I           =  (0.0,1.0)
      PI          =  3.1415926585
      PRINT 1000
      READ,       EPS
      EPS         =  - ABS(EPS)
      A           =  1.0 - EPS
      PRINT 1010
  100 READ,       THETA
      ZTILDE      =  A*CEXP(I*THETA*PI/180.) + EPS
      Z           =  ZTILDE + 1./ZTILDE
      ZPRIME      =  1. - 1./ZTILDE**2
      IF (CABS(ZPRIME) .LT. 1.E-10)      GO TO 200
      WPRIME      =  (1. - (A/(ZTILDE - EPS))**2)/ZPRIME
      CP          =  1. - CABS(WPRIME)**2
      PRINT 1020, Z,WPRIME,CP
      GO TO 100
  200 PRINT 1030
      GO TO 100
 1000 FORMAT(' INPUT THICKNESS PARAMETER EPS',)
 1010 FORMAT(' RESPOND TO QUESTION MARKS WITH THETA, IN DEGREES',
     +           //,7X,'X',9X,'Y',9X,'U',9X,'V',8X,'CP')
 1020 FORMAT(5F10.5)
 1030 FORMAT(' SINGULAR THETA')
      END
```